Legac-E Education

Loading an KSDS File

Although there had been previous examples of loading a Key Sequenced Data Set (KSDS) this version was produced in 2015 in support of exercises developed for a CICS Command Level programming course.

The program uses an input file containing 80-byte images in which the fields are blank separated. It converts these records to 72 byte records by removing the blanks.

The first 6-bytes of the input become the KEY fields for the output data set.

Included with the program listing which follows, is the copy book member which describes the input record, and the JCL associated with defining the cluster and executing the load program.

The program was updated in July 2018 to exploit dynamic file allocation for the KSDS Cluster. BPXWDYN is used to provide the dynamic allocation.

Legac-E Education

Loading an KSDS File

The Program

```
       IDENTIFICATION DIVISION.
       PROGRAM-ID.                    KSDSLOD.
       AUTHOR.                        T.R.SAMBROOKS.
           INSTALLATION.
           DATE-WRITTEN.              29TH AUG 2015.
       ENVIRONMENT DIVISION.
       CONFIGURATION SECTION.
      *-------------------------------------------------------------*
      * SAMPLE PROGRAM TO LOAD AN EMPTY VSAM KSDS DATA SET.         *
      *-------------------------------------------------------------*
       INPUT-OUTPUT SECTION.
       FILE-CONTROL.
           SELECT DSNS-IN             ASSIGN TO UT-S-KSDSDSNS.
           SELECT ENGINES-IN          ASSIGN TO UT-S-INDD.
           SELECT KSDS-FILE           ASSIGN TO INDEXED-OUTDD
                                      ORGANIZATION IS INDEXED
                                      ACCESS IS DYNAMIC
                                      RECORD KEY IS LOCO-KEY
                                      FILE STATUS     IS FSTAT-CODE
                                                         VSAM-CODE.
       DATA DIVISION.
       FILE SECTION.
       FD  DSNS-IN                    RECORDING MODE IS F
                                      LABEL RECORDS ARE STANDARD
                                      BLOCK CONTAINS 0 RECORDS
                                      RECORD CONTAINS 80 CHARACTERS
                                      DATA RECORD IS DSN-REC.
       01  DSN-REC                    PIC X(80).
       FD  ENGINES-IN                 RECORDING MODE IS F
                                      LABEL RECORDS ARE STANDARD
                                      BLOCK CONTAINS 0 RECORDS
                                      RECORD CONTAINS 80 CHARACTERS
                                      DATA RECORD IS FB-LOCO-REC.
       01  FB-LOCO-REC.
           COPY F100REC.
       FD  KSDS-FILE                  RECORD CONTAINS 72 CHARACTERS
                                      DATA RECORD IS KSDS-REC.
       01  KSDS-REC.
           COPY F72REC.
       WORKING-STORAGE SECTION.
       01  WS-ADHOC-CONSTANTS.
           03  SUB                    PIC S9(4) COMP VALUE +1.
           03  DYN-PGM                PIC X(8) VALUE 'BPXWDYN '.
           03  FSTAT-CODE             PIC XX.
           03  VSAM-CODE.
```

Legac-E Education

Loading an KSDS File

```
      05  R15-RETURN            PIC 99  COMP.
      05  VSAM-FUNCTION         PIC 9   COMP.
      05  VSAM-FEEDBACK         PIC 999 COMP.
    03  WS-EOF                  PIC X   VALUE 'M'.
      88  ALL-DONE              VALUE 'D'.
01  ALLOC-DYNFILE.
    03                           PIC S9(4) COMP VALUE +80.
    03                           PIC X(9) VALUE
       'ALLOC FI('.
    03  DYN-DDNA                PIC X(9) VALUE SPACES.
    03                           PIC X(17) VALUE
       ' SHR MSG(WTP) DA('.
    03  DYN-DSN                 PIC X(45) VALUE SPACES.
01  UNALLOC-DYNFILE.
    03                           PIC S9(4)  COMP VALUE  +27.
    03                           PIC X(8) VALUE
       'FREE FI('.
    03  DYN-DDNU                PIC X(8) VALUE SPACES.
    03                           PIC X(2) VALUE '  '.
    03                           PIC X(9) VALUE 'MSG(WTP) '.
PROCEDURE DIVISION.
A010-MAIN-PGM SECTION.
    PERFORM B010-INITIALIZE.
    PERFORM                      UNTIL ALL-DONE
      READ ENGINES-IN            AT END
                                    MOVE 'D' TO WS-EOF
           NOT AT END            MOVE FB-LOCO-KEY   TO LOCO-KEY
                                 MOVE FB-COY        TO COY
                                 MOVE FB-CAT-NO     TO CAT-NO
                                 MOVE FB-PRICE      TO PRICE
                                 MOVE FB-LOCO-DES   TO LOCO-DES
                               MOVE FB-LOCO-POWER TO LOCO-POWER
                               MOVE FB-LOCO-NAME  TO LOCO-NAME
                                 WRITE KSDS-REC

      END-READ
    END-PERFORM.
    PERFORM B020-TERMINATION.
A010-MAIN-PGM-EOJ.
    GOBACK.
```

Legac-E Education

Loading an KSDS File

```
B010-INITIALIZE.
     OPEN                          INPUT DSNS-IN.
     READ DSNS-IN                  AT END CLOSE DSNS-IN.
     UNSTRING DSN-REC              DELIMITED BY '='
                                   INTO DYN-DDNA, DYN-DSN.
     INSPECT DYN-DDNA              REPLACING FIRST ' ' BY ')'.
     INSPECT DYN-DSN               REPLACING FIRST ' ' BY ')'.
     CALL DYN-PGM                  USING ALLOC-DYNFILE.
     OPEN                          INPUT ENGINES-IN
                                   OUTPUT KSDS-FILE.
 B010-INITIALIZE-EXIT.
     EXIT.
 B020-TERMINATION.
       CLOSE                       ENGINES-IN
                                   KSDS-FILE.
       MOVE DYN-DDNA               TO DYN-DDNU.
       CALL DYN-PGM                USING UNALLOC-DYNFILE.
 B020B-TERMINATION-EXIT.
     EXIT.
*---------------------------------------------------------------*
* THIS IS BOTH THE LOGICAL AND PHYSICAL END OF - KSDSLOD        *
*---------------------------------------------------------------*
```

Input record Copy Book Member

```
      03  F100-RECORD.
          05  FB-LOCO-KEY.
              07  LOCO-TYPE       PIC X.
              07  LOCO-NO         PIC 9(5).
          05                       PIC X.
          05  FB-COY               PIC X(10).
          05                       PIC X.
          05  FB-CAT-NO           PIC X(7).
          05                       PIC X.
          05  FB-PRICE            PIC 9(3)V99.
          05                       PIC X.
          05  FB-LOCO-DES         PIC X(10).
          05                       PIC X.
          05  FB-LOCO-POWER       PIC X(5).
          05                       PIC X.
          05  FB-LOCO-NAME        PIC X(31).
```

Legac-E Education

Loading an KSDS File

Invoking JCL

```
//         EXPORT SYMLIST=STU
//         SET  STU=&SYSUID
//S0010    EXEC PGM=IDCAMS
//SYSPRINT DD   SYSOUT=*
//SYSIN    DD   *,SYMBOLS=JCLONLY
  DEL &STU..KSDS.FILE
  SET MAXCC=0
  DEF CLUSTER( NAME(&STU..KSDS.FILE)  -
      VOLUMES(*) -
      RECORDS(76 50)  -
      SHR(2 3)  -
      RECSZ(72 72)  -
      IXD  -
      KEYS(6 0) )  -
      DATA( NAME(&STU..KSDS.FILE.DATA))           -
      INDEX( NAME(&STU..KSDS.FILE.INDEX))
//         IF RC = 0 THEN
//S0020    EXEC PGM=KSDSLOD
//STEPLIB  DD   DISP=SHR,DSN=&STU..LOAD.LIBRARY
//INDD     DD   DISP=SHR,DSN=&STU..CICS.COB(ENGINES)
//KSDSDSNS DD   *,SYMBOLS=JCLONLY
OUTDD=&STU..KSDS.FILE
//         ENDIF
```